Linear Programming

Given the following information for a 1500 ac farm, construct a linear programming model that determines how much of each crop to plant.

Crop	Observed Acreage (ac)	Average Yield (bu/ac)	Price ($/bu)	Average costs ($/ac)
Wheat	500	42	$7.50	$192.0
Barley	200	70	$4.25	$169.5
Canola	450	38	$11.50	$229.0
Peas	250	45	$6.75	$163.8
Oats	100	110	$2.75	$152.50

1. Solve the following simple model using GAMS:

Maximize $\quad R = \sum_{k=1}^{n} (p_k x_k y_k - c_k x_k)$

Subject to $\quad \sum_{k=1}^{n} x_k \le 1500$

$x_k \ge 0$

2. Now include the following constraint and solve the problem again:

$$x_k \le x_k^{obs} + 0.01, \forall k$$

For these constraints find the associated shadow prices, λ_k, and use this information to modify the objective function above assuming a quadratic cost function: $c_k = a\, x_k + ½\, b\, x_k^2$. Then:

$b_k = 2 \times \lambda_k / x_k^{obs}$ and $a_k = c_k - ½ \times b_k \times x_k^{obs}$

Use the cost function in place of $c_k\, x_k$ in the objective function, so the revised objective is:

Maximize $\quad R = \sum_{k=1}^{n} (p_k x_k y_k - a_k x_k - \frac{1}{2} b_k x_k^2)$

Solve the revised problem using GAMS.

GAMS CODE

```
SETS
      crop    crops /wheat,barley,canola,peas,oats/

SCALAR     area  total cropland available to farmer /1500/;

TABLE input(crop, *) Input values
            obs    yld     price    cost
*        acres    bu/ac   $/bu     $/ac
wheat      500    42      7.50     192.0
barley     200    70      4.25     169.5
canola     450    38     11.50     229.0
peas       250    45      6.75     163.8
oats       100   110      2.75     152.5   ;

VARIABLES
       rev1      Objective for 1st stage of PMP
       rev2      Objective for 2nd stage of PMP
       x(crop)    Optimal number of acres in each crop    ;
POSITIVE VARIABLES  x ;

EQUATIONS
       obj1         Net revenue objective for 1st stage of PMP
       obj2         Net revenue objective for 2nd stage of PMP
       land         Total land constraint
       calib(crop)    Calibration constraints                          ;

obj1..     rev1 =E= sum(crop, x(crop)*(input(crop,'yld')*input(crop,'price') -  input(crop,'cost')));
land..     sum(crop, x(crop)) =L= area;
calib(crop)..   x(crop) =L= input(crop,'obs')+0.01;

MODEL pmp /obj1, land, calib/

SOLVE pmp using LP maximizing rev1;

* Assume TC = m0 x + 0.5 m1 x^2
PARAMETERS   m1(crop)  Slope parameter for MC curve
                    m0(crop)  Intercept parameter for MC curve        ;

m1(crop) = 2*calib.m(crop)/input(crop,'obs');
m0(crop) = input(crop,'cost') - 0.5*m1(crop)*input(crop,'obs');

obj2..    rev2 =E= sum(crop, input(crop,'price')*input(crop,'yld')*x(crop)
                     - m0(crop)*x(crop) - 0.5*m1(crop)*x(crop)*x(crop));

MODEL base /obj2, land/;

OPTION QCP=CPLEX;
SOLVE base using QCP maximizing rev2;
```